
News

Latest Version of Signal - Now With Dynamic Clamp
Our next release of Signal, which includes integrated dynamic clamp will be shown at the 
booth and the training days at this year's Society for Neuroscience Meeting in San Diego. If 
you are not attending the training days, please visit booth 2801 for a demonstration. 

Forum
The forum section on the CED web site now has over 350 registered members. This has 
been running since June 2008 and is becoming an increasingly useful resource with posts 
covering over 570 topics related to software and hardware, responded to by CED engineers 
and fellow users. If you register on the forum you can then use the search functions to locate 
useful posts relevant to your application as well as post your own questions which will be 
replied to promptly. You can also use the forum to put forward suggestions you feel would be 
useful additions to the Spike2 and Signal software.  

Future meetings and events
Society for Neuroscience 2010
San Diego Convention Center,
San Diego, CA, USA
November 13th – 17th  2010

If you are organising or attending a conference at which you feel CED would be interested in 
exhibiting please let us know about it by notifying sales@ced.co.uk

US Training Days at SfN
There are still a few places available at our US Training Days on the 11th and 12th of 
November at the Holiday Inn San Diego - On The Bay Hotel, prior to the Society for 
Neuroscience annual meeting. The training days are offered at $250 per day, including buffet 
lunch, materials and coffee and tea, and are suitable for both existing and prospective users 
of Spike2 and Signal. Places are filling up quickly so if you or any of your colleagues would 
like to attend, please follow the links to view the training day program and complete the 
registration form. Please give a short description of the type of research that you do and 
analysis that you are interested in. This will help us to tailor sessions for specific 
requirements.

Additional training opportunities
For those unable to attend the official US and UK training days it may be possible to arrange 
a small group or regional training session at your location. To discuss the options available 
please contact us. 

Latest versions of Spike2 and Signal
Updates for Spike2 and Signal are available from the CED downloads page, or by clicking on 
the links in the table below. Demonstration versions of the latest software are also available. 

 Spike2 downloads Released Signal downloads Released
Spike2 version 7.04 07/10 Signal version 4.0  8   08/10
Spike2 version 6.1  5  08/10 Signal version 3.13 05/10

Spike2 demo Signal demo

Q. I would like to keep a record of the template information associated with my wavemark 
channels. Is there a way to print or save these details?

A.  In both the New Wavemark and Edit Wavemark dialogs there is a Print option in 
the File menu. This enables you to generate a printout of the settings in the template 
parameters dialog together with images of the individual templates used for the 
channel. 

Typical printout of template information from Edit Wavemark dialog

Q. I have data recorded at a high sample rate which I would like to downsample before I 
export as text to put into a spreadsheet program as it will not currently fit into the 
available spreadsheet space. Is it possible to do this without writing a script? 

A. The virtual channel function available from version 4.06 allows you to create a channel 
with a sample rate either matched to the existing channels or specified independently. 
This allows you to create a channel at a lower sample rate than the disk based 
channels in the file and copy data into it from existing channels with the points 
interpolated by cubic splining. You can then copy or export this channel as normal.  

CED
Hint: If you right click in a data file and select the copy option to transfer Signal data into a spreadsheet package, using 'Paste' in the spreadsheet will add the data values in columns. 'Paste special' will also allow you to paste the data as an image in bitmap or metafile format.



When exporting or copying multiple channels, you must ensure you only copy the 
channels with matching sample rates as this function in Signal will copy the same 
number of points per channel.    

Virtual channel setup to copy an existing channel at a specified sample interval

Scripts: Spike2 

Q. Is it possible to classify spikes according to their latency from a stimulus event? When 
generating a PSTH from my wavemark channel  with raster plot enabled in the 
settings, l can clearly see a line of dots in the raster relating to a particular spike which 
I would like to isolate.  

A. The attached script Latency spike sort.s2s generates a PSTH showing the 
timing relationship between a trigger event channel and a spike wavemark channel. 
The raster data is extracted and displayed in an XY view and the user can then draw 
around areas in this plot and assign a code to points falling within the specified area. 
The script then finds the associated wavemarks in the data file and changes the 
marker codes accordingly.  

Once the file has been opened and the raster generated using the Get Raster 
option you can classify points within multiple sections. For each section, press Set 
Area then hold down the shift key and draw around an area with the left mouse button 
depressed. Once the button is released a dialog will appear prompting you to set the 
code number. Apply Changes then updates the data file with the assigned codes. 

N.B. The script requires Spike2 version 7 to run. 

Isolation of two spikes based on latency by drawing around points in a raster plot 

Back to top
Right-click the script icon and 
save to disk.
If you have any problems 
opening the embedded scripts in 
this newsletter please let us 

'$Latency spike sort|Script to classify wavemarks based on spike latency after stimulus

'Generates a PSTH and then copies raster daya to XY view

'User can then 'draw' around required events in raster plot and related spikes are coded accordingly



'Written by Steven Clifford 24/06/10



'The script is a 'work in progress' and is offered without guarantees.

'You must test it to see whether it is suitable for your application.





var data%;                                                      'Data file

var hist%;                                                      'PSTH from spike channel

var rast%;                                                      'XY view containing raster data

var spChan%;                                                    'Spike channel

var stChan%;                                                    'Stimulus channel

var spRastTime[1];                                              'Array of spike times relative to trigger

var spTimes[1];                                                 'Array of spike times

var spCodes%[1];                                                'Array of spike codes

var YVal[1];               

var nSpikes%;                                                   'Counter of spikes added to raster

var nTrigs%;                                                    'Number of trigger events 

var width := 0.1;               

var bSize := 0.001;             

var offset := 0;                

var drawCh%;                

var cols%[9];               

var c%;             

var code%[4];

var changeTimes[1];                                             'Array of spike times to reclassify

var changeCodes%[1];                                            'Array of code values to change

var nChange%;                                                   'Number of spikes to classify

var YStyle%;                                                    'Display of Y axis (sweep No. or time)





for c% := 0 to 8 do             

    cols%[c%] := Colour(c%+18);                                 'Get palette colours for wavemark codes

next;







ToolbarSet(1, "Quit");

ToolbarSet(3, "Open file", Open%);

ToolbarSet(5, "Get Raster", GetSpikes%);

ToolbarSet(7, "Set area", SetArea%);

ToolbarSet(8, "Apply changes", Apply%);

ToolbarEnable(5,0);

ToolbarEnable(7,0);

ToolbarEnable(8,0);

Toolbar("",1023);







Func Open%()

data% := FileOpen("",0,3);

if data% > 0 then

    Window(0,0,100,100);

    ToolbarSet(3, "Close file", Close%);

    ToolbarEnable(5,1);

endif;

return 1;

end;





Func Close%()

View(data%);

FileClose(0,-1);

data% := 0;

if rast% > 0 then

    View(rast%);

    FileClose(0,-1);

    rast% := 0;

endif;

ToolbarEnable(5,0);

ToolbarEnable(7,0);

ToolbarEnable(8,0);

ToolbarSet(3, "Open file", Open%);

return 1;

end;





Func GetSpikes%();

View(data%);

var ok%;

var int%;

var bins%;

var sTime,eTime;

var codeReset%;

DlgCreate("PSTH settings");

DlgChan(1, "Spike channel",16);

DlgChan(2, "Trigger channel",42);

DlgReal(3, "Width",0.001,10);

DlgReal(4, "Bin size",0.0001,1);

DlgReal(5, "Offset",0,10);

DlgList(6, "Y axis", "Sweep No.|Time",2);

ok% := DlgShow(spChan%,stChan%,width,bSize,offset,YStyle%);

if ok% > 0 then

    bins% := width/bSize;

    View(data%);

    CursorSet(2);

    int% := Interact("Position cursors around analysis time and press OK or select whole file",1023,0,"OK","Whole file");

    if int% = 1 then

        sTime := Cursor(1);

        eTime := Cursor(2);

    else

        sTime := 0;

        Cursor(1,0);

        eTime := MaxTime();

        Cursor(2,MaxTime());

    endif;

    codeReset% := Query("Reset existing codes?");

    if codeReset% > 0 then

        View(data%);

        code%[0]:= 0;

        MarkSet(spChan%,sTime,eTime,code%[]);

    endif;

    hist% := SetPSTH(spChan%,bins%,bSize,offset,stChan%,2);      'Generate PSTH with raster

    nTrigs% := Process(sTime,eTime,1,1);

    Window(0,0,50,100);

    FrontView(hist%);

    rast% := GetXYRaster%();

endif;    

return 1;

end;





Func GetXYRaster%()

var xy%;

var i%;

var sp%;                                                        'Number of spikes in sweep

var rasterChan% := 801;

var sweepSpikes[1000];                                          'Spike times for current sweep

var sT;                                                         'Trigger time

var swSpikes%;

var yDiff;                                                      'Y axis difference between min and max

xy% := FileNew(12,0);                                           'Set up new XY viw

XYSetChan(0,0,0,0);                                             'Create new channel (channel 2)

'Channel 2 is used for raster as this will be in front of channel 1 where the specified area is drawn.

'Area channel drawn with colour fill so must be behind raster channel

XYDrawMode(2,2,2);                                              'Set size of point in channel 2

XYJoin(1,4);                                                    'Set channel 1 to use show filled line

XYDrawMode(1,2,0);                                              'Make points invisible

XYColour(1,5,2);                                                'Set fill colour to light grey

XYDrawMode(-1,5,0);                                             'Turn off auto axis expansion if extra points added

XTitle$("Latency");

XUnits$("s");

View(hist%);



for i% := 1 to nTrigs% do                                       'Step through each triggered sweep in raster

    swSpikes% := RasterGet(rasterChan%,i%,sT,sweepSpikes[]);    'Get times of trigger and spikes

    if swSPikes% > 0 then                                       'If spikes found

        Resize spRastTime[nSpikes%+swSpikes%];                  'Resize arrays to total number of spikes 

        Resize spTimes[nSpikes%+swSpikes%];    

        Resize spCodes%[nSpikes%+swSpikes%];

        Resize YVal[nSpikes%+swSpikes%];

        ArrConst(spRastTime[nSpikes%:swSpikes%],sweepSpikes[:swSpikes%]);  'Copy current sweep spike times to array

        ArrSub(spRastTime[nSpikes%:swSpikes%],sT);                         'Subtract stimulus time

        ArrConst(spTimes[nSpikes%:swSpikes%],sweepSpikes[:swSpikes%]);     'Copy spike times to second array

        if YStyle% = 0 then

            ArrConst(YVal[nSpikes%:swSpikes%],i%);              'Set corresponding Y value for sweep points to plot in XY view

        else

            ArrConst(YVal[nSpikes%:swSpikes%],sT);

        endif;

        nSpikes% += swSpikes%;                                  'Add number of spikes to total

    endif;

next;

View(hist%);

FileClose(0,-1);

View(xy%);

View(xy%).XYAddData(2,spRastTime[],YVal[]);                     'Plot spike times in XY view (channel 2)

YDiff := (YVal[nSpikes%-1]-YVal[0])/20;                         'Calculate amount of Y axis to display around limits (range/20)

XRange(offset,offset+width);

YRange(1,YVal[0]-YDiff,YVal[nSpikes%-1]+YDiff);

WindowTitle$("Raster");

if YStyle% = 0 then

    ChanTitle$(0,"Sweep");

    ChanUnits$(0,"Number");

else

    ChanTitle$(0,"Time");

    ChanUnits$(0,"s");

endif;

View(data%).WIndow(50,0,100,100);

Window(0,0,50,100);

FrontView(xy%);

ToolbarEnable(7,1);

return xy%;

end;





Func SetArea%()                                                 'Function to specify area in XY view and classify spikes within

View(rast%);

XYDelete(1);                                                    'Delete any existing points in channel 1

ToolbarMouse(0,0,5,5,NChan%,up%,plotPos%);                      'Enable mouse functions

ToolbarText("Press Shift and draw around spikes to classify in raster");

return 1;

end;





Func nChan%(vh%,chan%,x,y,flags%)

'if vh% = rast% then

'XYDelete(1);

'endif;

return 1;

end;





Func Up%(vh%,chan%,x,y,flags%)

var fCode%;

var ok%;

var pts%;

var inIndexes%[1];

var i%;

var nCh%;                                                     'New channel to show specified points to change

DlgCreate("Classify spikes");

DlgInteger(1, "Set code",0,8);

ToolbarText("");

ok% := DlgShow(fCode%);

if ok% > 0 then                                               'if Ok pressed

    pts% := XYInChan(2,1);                                    'Count raster points within defined area

    if pts% > 0 then                                          'If any points present

        nCh% := XYSetChan(0,0,0,0,cols%[fCode%]);

        XYDrawMode(nCh%,1,4);

        XYDrawMode(nCh%,2,3);

        Resize inIndexes%[pts%];                              'Resize array

        pts% := XYInChan(2,1,inIndexes%[]);                   'Get indexes of points within area

        Resize changeTimes[nChange% + pts%];

        Resize changeCodes%[nChange% + pts%];

        for i% := 0 to pts%-1 do

            changeTimes[nChange% + i%] := spTimes[inIndexes%[i%]];

            changeCodes%[nChange% + i%] := fCode%;

            XYAddData(nCh%,spRastTime[inIndexes%[i%]],YVal[inIndexes%[i%]]);

        next;

        nChange% += pts%;

        ToolbarEnable(8,1);

    endif;

endif;

return 1;

end;





Func PlotPos%(vh%,chan%,x,y,flags%)

'if vh% = rast% then

XYAddData(1,x,y);

'endif;

return 1;

end;





Func Apply%()

var i%;                                                       'Function to set codes in data file as classified by raster

View(data%);

for i% := 1 to nChange% do                                    'Step through array of specified points

    code%[0] := changeCodes%[i%-1];                           'Set code value

    MarkEdit(spChan%,changeTimes[i%-1],code%[]);              'Edit marker at time

next;

return 1;

end



CED
Latency spike sort.s2s

CED
Hint: You can load frequently used scripts into the script bar for easy one-click operation. Right clicking in the toolbar or application background area will give easy access to the script bar list dialog where you can assign scripts to individual buttons on the script bar and set button titles and tooltip descriptions. 



Scripts: Signal 

Q: I record single channel patch clamp data and would like to generate stability plots of 
the open/closed times and amplitudes.  Is  this  function available in  Signal  or can I 
access the data using the script language to produce my own plots? 

A: There are currently no built-in stability plot functions but we do have script access to 
the open/closed time data to generate stability plots.

The attached script  StabilityPlot.sgs is currently under development and can 
generate a number of different stability plots of amplitude and open/closed durations. 
The script requires that an idealised trace is present in the data file and can work on 
single or across a range of data frames. multiple frame operation requires that the data 
is recorded in gap-free mode. 

Amplitude plots display the level of each open and closed event against either the 
number of the event from the start of the specified period or against a pseudo time  
measurement where short events (<1ms) are spaced apart in X by a unit of 1 or longer 
events (>1ms) spaced by the square root of the event duration in ms.

 Duration plots display the mean open time, mean closed time and pOpen value for a 
specified number of events against either the running total of events or the time at the 
end of each group. For example, you can average sets of 50 events stepped by 25 
events to smooth the result. The mean values can be plotted against the event count 
(ie 50, 75, 100 etc) or the time of the last event in each set of 50.

This script is still a work in progress so any feedback or suggestions for improvements 
are welcome. 

Single channel stability plots

Did you know…?

The data update mode in Signal can be set for the current data file in the File menu. This 
determines what happens to modified data (eg. filtered, smoothed, rectified) when you move 
away from the current frame. The options available are Discard changes, Query the 
user and Always save. Remember that saved changes cannot be undone so it is always 
worth saving a copy of the original data file if you are planning to keep modifications.

The default for this is set in the Data tab of the Preferences dialog under the Edit menu. 

Recent questions

Q. I have a waveform channel indicating times of activity and would like to generate a plot  
to show the activity duration for each event against the event number. 

'$StabilityPlot.sgs|Signal script to generate stability plots from single channel idealised trace channels







var ch% := 201;

var idChans%[100];                                                  'Array of idealised trace channels in file

var nEv%;

var data%;

var gapFree%,gotGFVar%;                                             'Flag for gap free sampled data file

var plTypes%[2];                                                    'Type of plot to generate (amplitude or duration)

var ampXType% := 0;                                                 'X axis display type for amplitude plots (event number or pseudo time)

var durXType% := 0;                                                 'X axis display type for duration plots (mean of set number of events or time)

var ampPlot%;                                                       'Amplitude plot handle   

var durPlot%;                                                       'Duration plot handle

var durAvPts% := 50;                                                'Points to average in duration plot

var durAvStep% := 25;                                               'Points to step between averages

var durAvTime := 1;                                                 'Time per average in duration plot

var avDlgItem%,stepDlgItem%,aTDlgItem%;                             'Item numbers for text displays in dialog to allow hiding

var sFrame%:= 1;                                                    'Start and end frame if multiple frames to be included

var eFrame% := 1;

var frames%[1];

var fName$;                                                         'Data file name (for use in stability plot title)

var startState%;                                                    'State of first event (open or closed)



var dataArr[2000][6];                                               'Array of details about each event

'[0] = event number

'[1] = event start time

'[2] = pseudo start time

'[3] = Event duration (ms)

'[4] = amplitude

'[5] = flag for open or closed

'[6] = Total flag value



var OpClArr[1][6];                                                  'Array of alternating open/closed events (amalgamated consecutive similar events)

var nAlt%;                                                          'Number of sorted alternating events after multiple open/closed events amalgamated





ToolbarSet(1,"Quit");

ToolbarSet(3, "Get file", GetFile%);

ToolbarSet(5, "Stability plot",DoPlot%);

ToolbarEnable(5,0);

Toolbar("Single channel stability plot",1023);







Func GetFile%()

var useCur%;

if ViewKind(FrontView())= 0 then

    useCur% := Query("Use current file");

    if useCur% > 0 then

        data% := FrontView();

    else

        data% := FileOpen("",0,2);

    endif;

else

    data% := FileOpen("",0,2);

endif;

Window(0,0,100,100);

FrontView(data%);

if data% > 0 then

    ChanList(idChans%[],4);                                           'Get list of idealised trace channels

    if idChans%[0] = 0 then                                           'Stop if no suitable channels present

        Message("         File contains no idealised trace channels\nPlease generate idealised trace before generating stability plots");

    else

        ToolbarEnable(5,1);

    endif;

    gapFree% := FileGetIntVar("FFlags",gotGfVar%);                    'Find if file recorded in gap free mode

    fName$ := FileName$(3) + FileName$(4);

endif;

return 1;

end;





Func DoPlot%()

var ok%;

var i%;

var tEv%;                                                             'Running total of events over multiple frames

var pTEv%;                                                            'Total at end of previous frame

var cTime;                                                            'Cumulative time

View(data%);

DlgCreate("Stability plot setup",0,0,50);

DlgChan(1, "Channel to analyse",4);

DlgGroup("Required plots",1,4,48,8.5);

DlgGroup("Amplitude",2,5,46,2.5);

DlgCheck(2, "",6,6);

DlgGroup("Duration",2,8,46,4);

DlgCheck(3, "",6,9);

DlgList(4, 14,"Pseudo time|Event number",2,33,6);

DlgText("X axis",20,6);

DlgText("X axis",20,9);

DlgList(5, 15,"Event number|Time",2,32,9);

avDlgItem% := DlgText("Points per average",3,10.5);

DlgInteger(6,5,1,1000,20,10.5);

stepDlgItem% := DlgText("Points to step",29,10.5);

DlgInteger(7,5,1,1000,42,10.5);

'aTDlgItem% := DlgText("Time to average(s)",3,10.5);

'DlgReal(8,5,0.01,1000,42,10.5);

if gapFree% > 0 then

    DlgText("Start frame",2,2.5);

    DlgInteger(8, 5,1,FrameCount(),15,2.5);

    DlgText("End frame",30,2.5);

    DlgInteger(9, 5, 1, FrameCount(),42,2.5);

else

    DlgText("File not gap free so only current frame will be included",2,2.5);

endif;

DlgAllow(1023,0,changed%);

if gapFree% > 0 then

    ok% := DlgShow(ch%,plTypes%[],ampXType%,durXType%,durAvPts%,durAvStep%,sFrame%,eFrame%);

else

    ok% := DlgShow(ch%,plTypes%[],ampXType%,durXType%,durAvPts%,durAvStep%);

endif;



if ok% > 0 then

    View(data%);

    if gapFree% = 0 then

        nEv% := ChanMeasure(ch%,15,minTime(),maxTime());            'Count events in idealised trace 

        tEv% := nEv%;

        sFrame% := Frame();

        eFrame% := Frame();

        'Message(nEv%);

        'Resize dataArr[nEv%][6]; 

        AddOpClData%(ch%,nEv%,tEv%,0,0,Frame());

    else

        cTime := 0;

        Resize frames%[(eFrame%-sFrame%)+2];                        'Get list of frames to inlcude in array

        frames%[0] := (eFrame%-sFrame%)+1;

        ArrConst(frames%[1:],1);

        ArrIntgl(frames%[1:]);

        ArrAdd(frames%[1:],sFrame%-1);

        for i% := frames%[1] to frames%[frames%[0]] do

            Frame(i%);

            'Yield(0.2);                                            'For testing purposes

            nEv% := ChanMeasure(ch%,15,minTime(),maxTime());        'Count events in idealised trace

            if i% < eFrame% then nEv% += 1;endif;                   'Need to add an additional event if not the last frame

            tEv% += nEv%;                                           'Add event count to running total   

            AddOPCLData%(ch%,nEv%,tEv%,pTEv%,cTime,i%);

            pTEv% := tEv%;

            cTime += MaxTime()*1000.0;                              'Add frame length to cumulative time

            next;

    endif;

    nAlt% := GetAltDataArr(TEv%,dataArr[][],OpClArr[][]);

    GetOpClArr%(nAlt%,OpClArr[][]);

endif;

return 1;

end;





Func Changed%(item%);                                                'Function to enable/disable and show/hide appropriate dialog items 

DlgEnable(DlgValue(2),4);

DlgEnable(DlgValue(3),5);

DlgEnable(DlgValue(3),avDlgItem%);

DlgEnable(DlgValue(3),stepDlgItem%);

DlgEnable(DlgValue(3),6);

DlgEnable(DlgValue(3),7);

'DlgEnable(DlgValue(3),aTDlgItem%);

'DlgEnable(DlgValue(3),8);

DlgEnable(DlgValue(2)+DlgValue(3),-1);

return 1;

end;



Func AddOPCLData%(c%,n%,t%,p%,ct,fr%)                       'Function to get idealised trace event information for current frame

'c% = channel

'n% = Number of events in frame

't% = Cunulative event count at end of frame 

'p% = number of events prior to this frame

'ct = cumulative time

'fr% = frame number

var i%;

var evTime;

var evTimeMs;

var per;

var flags%;

var amp;

var PseudInc;

var meanAmp;



Resize dataArr[t%][6];

if fr% > sFrame% then                                               'If not the first frame we ned to combine start event with last event in previous frame

    evTime := MinTime();                                

    OpClEventGet(c%,2,evTime,per,amp,flags%);                       'Get duration and amplitude of first event

    evTimeMs := evTime * 1000.0;                                    'Convert event time to ms

    meanAmp := ((dataArr[p%-1][4]*dataArr[p%-1][3]) + ((per * 1000.0)*amp)) / (dataArr[p%-1][3]+ (per * 1000.0));   'Get weighted mean amplitude for total event duration

    'PrintLog("%f\t%f\t%f\t%f",dataArr[p%-1][4],dataArr[p%-1][3],per * 1000.0,amp);                                  'For testing purposes

    dataArr[p%-1][4] := meanAmp;                                    'Set mean amp for last evnt in previous frame

    dataArr[p%-1][3] += per * 1000.0;                               'Add current event length to last in previous frame

    'PrintLog("\t%f\t%f\n",dataArr[p%-1][4],dataArr[p%-1][3]);                                                       'For testing purposes

else

    OpClEventGet(c%,3,evTime,per,amp,flags%);

    if flags% band 128 then                                         'Get state of first event

        startState%:= 0

    else

        startState% := 1;

    endif;       

endif;    

for i% := 1 to n% do

    OpClEventGet(c%,3,evTime,per,amp,flags%);

    if flags% band 128 then                                          'Get open or closed state flag

        dataArr[i%+p%-1][5] := 0;

    else

        dataArr[i%+p%-1][5] := 1;

    endif;

    evTimeMs := ct + (evTime * 1000.0);                             'Convert event time to ms

    dataArr[i%+p%-1][1] := evTimeMs;                                'Add to array

    'evTime += per;                                                 'Get time at end of event

    dataArr[i%+p%-1][0] := p% + i%;                                 'Enter event number into array

    dataArr[i%+p%-1][4]:= amp;                                      'Enter amplitude into array

    dataArr[i%+p%-1][3] := per * 1000.0;                            'Enter duration in ms into array

    next;

return 1;

end;











Func GetOpClArr%(n%,OpClArr[][])

var i%,j%;

var pseudinc;

var openArr[(n%/2)+1][6];                                           'Array of details for just open events

var closedArr[(n%/2)+1][6];                                         'closed

var nOpen% := 0;                                                    'Counters for number of open and closed events

var nClosed% := 0;





'ArrConst(pointArr[],1);

'ArrIntgl(pointArr[]);

for i% := 1 to n% do

    if opClArr[i%-1][3] <= 1 then                                   'Get pseudo time (use 1 if < 1ms or sqrt of time if > 1ms). 

        opClArr[i%-1][2] := 1;

    else

        pseudInc := opClArr[i%-1][3];

        opClArr[i%-1][2] := Sqrt(pseudInc);

    endif;

    if opClArr[i%-1][5] = 1 then                                    'Enter results into separate array for open events

        Resize openArr[nOpen%+1][6];

        openArr[nOpen%][0] := i%;

        openArr[nOpen%][1] := opClArr[i%-1][1];

        openArr[nOpen%][2] := ArrSum(opClArr[:i%][2]);

        for j% := 3 to 5 do

            openArr[nOpen%][j%] := opClArr[i%-1][j%];

            next;

        nOpen% += 1;

    else                                                            'And closed events

        Resize closedArr[nClosed%+1][6];

        closedArr[nClosed%][0] := i%;

        closedArr[nclosed%][1] := opClArr[i%-1][1];

        closedArr[nclosed%][2] := ArrSum(opClArr[:i%][2]);

        for j% := 3 to 5 do

            closedArr[nClosed%][j%] := opClArr[i%-1][j%];

            next;

        nClosed% += 1;

    endif;

    

    'dataArr[i%-1][5] := flags%;

    'PrintLog("%d\n",flags%);

    next;

ArrIntgl(opClArr[][2]);                                             'Integrate pseudo time interval values

View(data%);

PrintLog("%f\n%f",MaxTime(),(dataArr[n%-1][1] + dataArr[n%-1][3])/1000.0);

if plTypes%[0] > 0 then

    ampPlot% := PlotAmp%(nOpen%,nClosed%,openArr[][],closedArr[][])

endif;

if plTypes%[1] > 0 then

    durPlot% := PlotDur%(nOpen%,nClosed%,openArr[][],closedArr[][],durXType%,durAvPts%,durAvStep%,durAvTime);

endif;

return 1;

end;







Func PlotAmp%(op%,cl%,oArr[][],clArr[][])

var plot%;

var pTitle$;                                                                    'Stability plot title

plot% := FileNew(12);                                                           'Open new XY document

XYSetChan(1,0,0,0,16);                                                          'Set up channels

ChanTitle$(1, "Open");

XYSetChan(0,0,0,0,24);

ChanTitle$(2, "Closed");

ChanUnits$(1,View(data%).ChanUnits$(ch%));

'XYAddData(1,dataArr[][1],dataArr[][4]);                'Plot data to XY view

if ampXType% = 0 then                                                           'If plotting pseudo time on X axis

    pTitle$ := Print$("Pseudo time v amplitude stability plot for %s",fName$);

    WindowTitle$(pTitle$);

    XTitle$("");

    XUnits$("Time");                                                            'Label X axis

    'ArrDiv(oArr[:op%][1],1000.0);                                             

    'ArrDiv(clArr[:cl%][1],1000.0);                                            

    XYAddData(1,oArr[:op%][2],oArr[:op%][4]);                                  'Plot pseudo time on X against open amp on Y

    XYAddData(2,clArr[:cl%][2],clArr[:cl%][4]);                                '                             closed

else                                                                           'If plotting event number

    pTitle$ := Print$("Event number v amplitude stability plot for %s",fName$);

    WindowTitle$(pTitle$);

    XTitle$("");

    XUnits$("Event");                                                          'Label x axis

    XYAddData(1,oArr[:op%][0],oArr[:op%][4]);                                  'Plot event number against open amp

    XYAddData(2,clArr[:cl%][0],clArr[:cl%][4]);                                '                         closed   

endif;

XYDrawMode(1,2,3);                                                             'Set dot size for channels to 3

XYDrawMode(2,2,3);

XYKey(1,1);                                                                    'Show key

XYKey(2,1);                                                                    'Make key transparent

if ArrSum(plTypes%[]) > 1 then                                                 'If plotting amp and duration

    Window(0,0,50,100);                                                        'Position on left side of Signal window

else

    Window(0,0,100,100);                                                       'Else fill Signal window

endif;

FrontView(plot%);

return plot%;

end;





Func PlotDur%(nOp%,nCl%,openArr[][],closedArr[][],x%,avPts%,avStep%,avTime)

var plot%;

var pTitle$;                                                                   'Stability plot title

var nSecs%;

var i%;

var sIndex% := 0;

var eIndex%;

var opChan%,clChan%,pOpenChan%;                                                'XY view channel numbers

var nOpSec%,nClSec%;                                                           'Number of open and closed sections to plot

nOpSec% := Trunc((nOp%-avPts%)/avStep%)+1;                                     'Get number of mean open points

nClSec% := Trunc((nCl%-avPts%)/avStep%)+1;                             

if nOpSec% < nClSec% then                                                      'Ensure we have the same number of open and closed section

    nSecs% := NOpSec%;                             

else                               

    nSecs% := nClSec%;                             

endif;                             

var durArr[nSecs%][8];                                                         'Array of results for each event number step

for i% := 1 to NSecs% do                                                       'For each set of events

    eIndex% := sIndex% + avPts%;

    durArr[i%-1][0] := ArrSum(openArr[sIndex%:avPts%][3],durArr[i%-1][1]);     'Get total and mean duration of open times

    durArr[i%-1][2] := ArrSum(closedArr[sIndex%:avPts%][3],durArr[i%-1][3]);   'Get total and mean duration of closed times

    durArr[i%-1][4] := durArr[i%-1][0] + durArr[i%-1][2];                      'Get total time

    durArr[i%-1][5] := durArr[i%-1][0] / durArr[i%-1][4];                      'Get open probability (pOpen)

    durArr[i%-1][6] := eIndex%;                                                'Get point number at end of section

    if startState% = 0 then                                                    'If first event in file is closed

        durArr[i%-1][7] := openArr[eIndex%][1];                                'Set time to last open time in section

    else

        durArr[i%-1][7] := closedArr[eIndex%][1];                              'or last closed if first event open

    endif;

    sIndex% += avStep%;

    next;

plot% := FileNew(12);

pTitle$ := Print$("Event number v duration stability plot for %s",fName$);

WindowTitle$(pTitle$);

opChan% := 1;

XYSetChan(opChan%,0,0,0,14);

clChan% := XYSetChan(0,0,0,0,15);

pOpenChan% := XYSetChan(0,0,0,0,16);

XYDrawMode(opChan%,2,3);

XYDrawMode(clChan%,2,3);

XYDrawMode(pOpenChan%,2,3);

ChanTitle$(opChan%,"Mean open time");

ChanTitle$(clChan%,"Mean closed time");

ChanTitle$(pOpenChan%,"pOpen");

if x% = 0 then

    XYAddData(opChan%,durArr[][6],durArr[][1]);                                 'Add open durations to XY plot

    XYAddData(clChan%,durArr[][6],durArr[][3]);                                 '    closed                   

    XYAddData(pOpenChan%,durArr[][6],durArr[][5]);                              'Add pOpen values to XY plot

    XUnits$("Event");

else

    XYAddData(opChan%,durArr[][7],durArr[][1]);                                 'Add open durations to XY plot

    XYAddData(clChan%,durArr[][7],durArr[][3]);                                 '    closed                   

    XYAddData(pOpenChan%,durArr[][7],durArr[][5]);                              'Add pOpen values to XY plot

    XUnits$("Time");

endif;

XYKey(1,1);                                                                     'Show key

XYKey(2,1);                                                                     'Make key transparent

XTitle$("");

ChanUnits$(1, "ms");

YAxisAttrib(-1,1);                                                              'Show Y axis in log mode

Optimise(-1);                                                                   'Optimise Y axis

if ArrSum(plTypes%[]) > 1 then                                                  'Position window based on number of plots (right half or fill Signal window)

    Window(50,0,100,100);

else

    Window(0,0,100,100);

endif;

FrontView(plot%);

return plot%;

end;







Func GetAltDataArr(e%,Arr[][],&aArr[][]);

var i%;                                                                             'Event number

var evNum%;                                                                         'Amalgamated event number

var evTime;                                                                         'Array containing current similar event durations

var evLevs[100];                                                                    'Current similar event amplitudes

var sTime;                                                                          'Start time for current event

var tTime;                                                                          'Total time for current event

var mAmp;                                                                           'mean amplitude for current event

var nEvFlag% := 1;                                                                  'Flag for starting new amalgamated event

var cCount% := 0;                               

'ArrConst(aArr[0][],Arr[0][]);                                                      'Copy first event data

'Message("%f",Arr[1][1]);                               

'tTime := Arr[1][1];                                

'Message("%f",tTime);                               

sTime := Arr[0][1];                                                                 'Get start time of first event

'mAmp := tTime*Arr[0][4];                               

for i% := 1 to e%-1 do                              

    if Arr[i%][5] = Arr[i%-1][5] then                                               'If same state as previous event

        tTime += Arr[i%-1][3];                                                      'Add duration to total time

        mAmp += Arr[i%-1][4]*(Arr[i%][1]-Arr[i%-1][1]); '

        cCount% += 1;

    else

        Resize aArr[evNum%+1][6];

        tTime += Arr[i%-1][3];                                                      'Add duration to total time

        mAmp += Arr[i%-1][4]*(Arr[i%][1]-Arr[i%-1][1]); 

        aArr[evNum%][0] := evNum%;

        'aArr[evNum%][1] := Arr[i%-1][1];

        aArr[evNum%][1] := sTime;

        aArr[evNum%][3] := tTime;

        aArr[evNum%][4] := mAmp/tTime;

        aArr[evNum%][5] := Arr[i%][5];

        tTime := 0;

        mAMp := 0;

        cCount% := 0;

        evNum% += 1;

        if cCount% = 0 then

            sTime := Arr[i%][1];

        endif;

        

    endif;

    next;

ArrDiv(aArr[][1],1000.0);

'PlotOnOff%(aArr[:evNum%][]);                                                       'Generates XY plot of alternate on off events for testing purposes

return evNum%;

end;







Func PlotOnOff%(xy[][])

var onoff%;

View(data%);

onoff% := FileNew(12);

XYAddData(1,xy[][1],xy[][4]);

FrontView(onoff%);

return 1;

end;

A. The way to approach this would be to locate the start of each event first into a memory 
channel and then set cursor 0 to step through each detected point with another cursor 
locating the end of each event.  You could then generate an XY plot of the event count 
on X against the time between the two cursors on Y. 

The first step is to locate the start positions. From the Memory Buffer section of the 
analysis menu create a new event type memory channel. Import rising threshold 
times at a suitable amplitude. You can set a minimum interval at this point to reject 
events starting too close together. 

The images below illustrate typical settings for the displayed data file with the resultant 
XY plot of event number against duration. 

`

XY plot settings dialog

Original activity channel and resultant XY plot

